



Smart Contract Security Audit Report



The SlowMist Security Team received the team's application for smart contract security audit of the Coin98 Dollar (CUSD) on 2022.02.16. The following are the details and results of this smart contract security audit:

Token Name :

Coin98 Dollar (CUSD)

The contract address :

<https://etherscan.io/token/0xC285B7E09A4584D027E5BC36571785B515898246>

The audit items and results :

(Other unknown security vulnerabilities are not included in the audit responsibility scope)

NO.	Audit Items	Result
1	Replay Vulnerability	Passed
2	Denial of Service Vulnerability	Passed
3	Race Conditions Vulnerability	Passed
4	Authority Control Vulnerability	Some Risks
5	Integer Overflow and Underflow Vulnerability	Passed
6	Gas Optimization Audit	Passed
7	Design Logic Audit	Passed
8	Uninitialized Storage Pointers Vulnerability	Passed
9	Arithmetic Accuracy Deviation Vulnerability	Passed
10	"False top-up" Vulnerability	Passed
11	Malicious Event Log Audit	Passed
12	Scoping and Declarations Audit	Passed

NO.	Audit Items	Result
13	Safety Design Audit	Passed
14	Non-privacy/Non-dark Coin Audit	Passed

Audit Result : Low Risk

Audit Number : 0X002202210003

Audit Date : 2022.02.16 - 2022.02.21

Audit Team : SlowMist Security Team

Summary conclusion : This is a token contract that does not contain the tokenVault section. The total amount of contract tokens can be changed, users can burn their own tokens through the burn function. SafeMath security module is used, which is a recommended approach. The contract does not have the Overflow and the Race Conditions issue.

During the audit, we found the following issue:

1. The owner role can add a new minter and the minter role can mint tokens arbitrarily through the mint function and there is no upper limit on the amount of tokens that can be minted that will lead to token inflation. After communication with the project team, the project team expresses that the mint function is required in their contract because they need it to work in other contract. The other contract will lock the stablecoins from the user then mint new CUSD based on a 1:1 ratio. Besides, the owner will be controlled by multisig.
2. The owner role can withdraw any tokens from this contract to a specific address.
3. There is no upper limit of the gas to execute the low-level call to withdraw the native tokens.

The source code:

```
/**
 *Submitted for verification at Etherscan.io on 2022-07-16
 */
```

```
// SPDX-License-Identifier: MIT
//SlowMist// The contract does not have the Overflow and the Race Conditions issue
pragma solidity ^0.8.0;

/**
 * @dev Wrappers over Solidity's arithmetic operations with added overflow
 * checks.
 *
 * Arithmetic operations in Solidity wrap on overflow. This can easily result
 * in bugs, because programmers usually assume that an overflow raises an
 * error, which is the standard behavior in high level programming languages.
 * `SafeMath` restores this intuition by reverting the transaction when an
 * operation overflows.
 *
 * Using this library instead of the unchecked operations eliminates an entire
 * class of bugs, so it's recommended to use it always.
 */
// CAUTION
// This version of SafeMath should only be used with Solidity 0.8 or later,
// because it relies on the compiler's built in overflow checks.

/**
 * @dev Wrappers over Solidity's arithmetic operations.
 *
 * NOTE: `SafeMath` is generally not needed starting with Solidity 0.8, since the
 * compiler
 * now has built in overflow checking.
 */
//SlowMist// OpenZeppelin's SafeMath security module is used, which is a recommended
//approach
library SafeMath {
    /**
     * @dev Returns the addition of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function tryAdd(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            uint256 c = a + b;
            if (c < a) return (false, 0);
            return (true, c);
        }
    }
}
```

```

/**
 * @dev Returns the subtraction of two unsigned integers, with an overflow flag.
 *
 * _Available since v3.4._
 */
function trySub(uint256 a, uint256 b) internal pure returns (bool, uint256) {
    unchecked {
        if (b > a) return (false, 0);
        return (true, a - b);
    }
}

/**
 * @dev Returns the multiplication of two unsigned integers, with an overflow flag.
 *
 * _Available since v3.4._
 */
function tryMul(uint256 a, uint256 b) internal pure returns (bool, uint256) {
    unchecked {
        // Gas optimization: this is cheaper than requiring 'a' not being zero, but
the
        // benefit is lost if 'b' is also tested.
        // See: https://github.com/OpenZeppelin/openzeppelin-contracts/pull/522
        if (a == 0) return (true, 0);
        uint256 c = a * b;
        if (c / a != b) return (false, 0);
        return (true, c);
    }
}

/**
 * @dev Returns the division of two unsigned integers, with a division by zero
flag.
 *
 * _Available since v3.4._
 */
function tryDiv(uint256 a, uint256 b) internal pure returns (bool, uint256) {
    unchecked {
        if (b == 0) return (false, 0);
        return (true, a / b);
    }
}

```

```
/**
 * @dev Returns the remainder of dividing two unsigned integers, with a division by
zero flag.
 *
 * _Available since v3.4._
 */
function tryMod(uint256 a, uint256 b) internal pure returns (bool, uint256) {
    unchecked {
        if (b == 0) return (false, 0);
        return (true, a % b);
    }
}

/**
 * @dev Returns the addition of two unsigned integers, reverting on
 * overflow.
 *
 * Counterpart to Solidity's `+` operator.
 *
 * Requirements:
 *
 * - Addition cannot overflow.
 */
function add(uint256 a, uint256 b) internal pure returns (uint256) {
    return a + b;
}

/**
 * @dev Returns the subtraction of two unsigned integers, reverting on
 * overflow (when the result is negative).
 *
 * Counterpart to Solidity's `-` operator.
 *
 * Requirements:
 *
 * - Subtraction cannot overflow.
 */
function sub(uint256 a, uint256 b) internal pure returns (uint256) {
    return a - b;
}

/**
 * @dev Returns the multiplication of two unsigned integers, reverting on
 * overflow.
```

```

*
* Counterpart to Solidity's `*` operator.
*
* Requirements:
*
* - Multiplication cannot overflow.
*/
function mul(uint256 a, uint256 b) internal pure returns (uint256) {
    return a * b;
}

/**
 * @dev Returns the integer division of two unsigned integers, reverting on
 * division by zero. The result is rounded towards zero.
 *
 * Counterpart to Solidity's `/` operator.
 *
 * Requirements:
 *
 * - The divisor cannot be zero.
 */
function div(uint256 a, uint256 b) internal pure returns (uint256) {
    return a / b;
}

/**
 * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer
modulo),
 * reverting when dividing by zero.
 *
 * Counterpart to Solidity's `%` operator. This function uses a `revert`
 * opcode (which leaves remaining gas untouched) while Solidity uses an
 * invalid opcode to revert (consuming all remaining gas).
 *
 * Requirements:
 *
 * - The divisor cannot be zero.
 */
function mod(uint256 a, uint256 b) internal pure returns (uint256) {
    return a % b;
}

/**
 * @dev Returns the subtraction of two unsigned integers, reverting with custom

```

```

message on
    * overflow (when the result is negative).
    *
    * CAUTION: This function is deprecated because it requires allocating memory for
the error
    * message unnecessarily. For custom revert reasons use {trySub}.
    *
    * Counterpart to Solidity's `-` operator.
    *
    * Requirements:
    *
    * - Subtraction cannot overflow.
    */
function sub(
    uint256 a,
    uint256 b,
    string memory errorMessage
) internal pure returns (uint256) {
    unchecked {
        require(b <= a, errorMessage);
        return a - b;
    }
}

/**
    * @dev Returns the integer division of two unsigned integers, reverting with
custom message on
    * division by zero. The result is rounded towards zero.
    *
    * Counterpart to Solidity's `/` operator. Note: this function uses a
    * `revert` opcode (which leaves remaining gas untouched) while Solidity
    * uses an invalid opcode to revert (consuming all remaining gas).
    *
    * Requirements:
    *
    * - The divisor cannot be zero.
    */
function div(
    uint256 a,
    uint256 b,
    string memory errorMessage
) internal pure returns (uint256) {
    unchecked {
        require(b > 0, errorMessage);

```



```

        return a / b;
    }
}

/**
 * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer
modulo),
 * reverting with custom message when dividing by zero.
 *
 * CAUTION: This function is deprecated because it requires allocating memory for
the error
 * message unnecessarily. For custom revert reasons use {tryMod}.
 *
 * Counterpart to Solidity's `%` operator. This function uses a `revert`
 * opcode (which leaves remaining gas untouched) while Solidity uses an
 * invalid opcode to revert (consuming all remaining gas).
 *
 * Requirements:
 *
 * - The divisor cannot be zero.
 */
function mod(
    uint256 a,
    uint256 b,
    string memory errorMessage
) internal pure returns (uint256) {
    unchecked {
        require(b > 0, errorMessage);
        return a % b;
    }
}

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.

```

```

*/
function balanceOf(address account) external view returns (uint256);

/**
 * @dev Moves `amount` tokens from the caller's account to `recipient`.
 *
 * Returns a boolean value indicating whether the operation succeeded.
 *
 * Emits a {Transfer} event.
 */
function transfer(address recipient, uint256 amount) external returns (bool);

/**
 * @dev Returns the remaining number of tokens that `spender` will be
 * allowed to spend on behalf of `owner` through {transferFrom}. This is
 * zero by default.
 *
 * This value changes when {approve} or {transferFrom} are called.
 */
function allowance(address owner, address spender) external view returns (uint256);

/**
 * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
 *
 * Returns a boolean value indicating whether the operation succeeded.
 *
 * IMPORTANT: Beware that changing an allowance with this method brings the risk
 * that someone may use both the old and the new allowance by unfortunate
 * transaction ordering. One possible solution to mitigate this race
 * condition is to first reduce the spender's allowance to 0 and set the
 * desired value afterwards:
 * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
 *
 * Emits an {Approval} event.
 */
function approve(address spender, uint256 amount) external returns (bool);

/**
 * @dev Moves `amount` tokens from `sender` to `recipient` using the
 * allowance mechanism. `amount` is then deducted from the caller's
 * allowance.
 *
 * Returns a boolean value indicating whether the operation succeeded.
 */

```

```

    * Emits a {Transfer} event.
    */
    function transferFrom(address sender, address recipient, uint256 amount) external
returns (bool);

/**
 * @dev Emitted when `value` tokens are moved from one account (`from`) to
 * another (`to`).
 *
 * Note that `value` may be zero.
 */
event Transfer(address indexed from, address indexed to, uint256 value);

/**
 * @dev Emitted when the allowance of a `spender` for an `owner` is set by
 * a call to {approve}. `value` is the new allowance.
 */
event Approval(address indexed owner, address indexed spender, uint256 value);
}

/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721 {

    /**
     * @dev Transfers `tokenId` token from `from` to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {safeTransferFrom} whenever
    possible.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either
    {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,

```

```

    uint256 tokenId
) external;
}

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     * - an externally-owned account
     * - a contract in construction
     * - an address where a contract will be created
     * - an address where a contract lived, but was destroyed
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability,
     breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by
     calling from a contract
     * constructor.
     *
     * [IMPORTANT]
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }
}

```

```

/**
 * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
 * `recipient`, forwarding all available gas and reverting on errors.
 *
 * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
 * of certain opcodes, possibly making contracts go over the 2300 gas limit
 * imposed by `transfer`, making them unable to receive funds via
 * `transfer`. {sendValue} removes this limitation.
 *
 * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-
now/[Learn more].
 *
 * IMPORTANT: because control is transferred to `recipient`, care must be
 * taken to not create reentrancy vulnerabilities. Consider using
 * {ReentrancyGuard} or the
 * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-
checks-effects-interactions-pattern[checks-effects-interactions pattern].
 */
function sendValue(address payable recipient, uint256 amount) internal {
    require(address(this).balance >= amount, "Address: insufficient balance");

    (bool success, ) = recipient.call{value: amount}("");
    require(success, "Address: unable to send value, recipient may have reverted");
}

/**
 * @dev Performs a Solidity function call using a low level `call`. A
 * plain `call` is an unsafe replacement for a function call: use this
 * function instead.
 *
 * If `target` reverts with a revert reason, it is bubbled up by this
 * function (like regular Solidity function calls).
 *
 * Returns the raw returned data. To convert to the expected return value,
 * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?
highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
 *
 * Requirements:
 *
 * - `target` must be a contract.
 * - calling `target` with `data` must not revert.
 *
 * _Available since v3.1._
 */

```

```

function functionCall(address target, bytes memory data) internal returns (bytes
memory) {
    return functionCall(target, data, "Address: low-level call failed");
}

/**
 * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but
with
 * `errorMessage` as a fallback revert reason when `target` reverts.
 *
 * _Available since v3.1._
 */
function functionCall(
    address target,
    bytes memory data,
    string memory errorMessage
) internal returns (bytes memory) {
    return functionCallWithValue(target, data, 0, errorMessage);
}

/**
 * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
 * but also transferring `value` wei to `target`.
 *
 * Requirements:
 *
 * - the calling contract must have an ETH balance of at least `value`.
 * - the called Solidity function must be `payable`.
 *
 * _Available since v3.1._
 */
function functionCallWithValue(
    address target,
    bytes memory data,
    uint256 value
) internal returns (bytes memory) {
    return functionCallWithValue(target, data, value, "Address: low-level call with
value failed");
}

/**
 * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}
[`functionCallWithValue`], but
 * with `errorMessage` as a fallback revert reason when `target` reverts.

```

```

*
* _Available since v3.1._
*/
function functionCallWithValue(
    address target,
    bytes memory data,
    uint256 value,
    string memory errorMessage
) internal returns (bytes memory) {
    require(address(this).balance >= value, "Address: insufficient balance for
call");
    require(isContract(target), "Address: call to non-contract");

    (bool success, bytes memory returndata) = target.call{value: value}(data);
    return verifyCallResult(success, returndata, errorMessage);
}

/**
 * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
 * but performing a static call.
 *
 * _Available since v3.3._
 */
function functionStaticCall(address target, bytes memory data) internal view
returns (bytes memory) {
    return functionStaticCall(target, data, "Address: low-level static call
failed");
}

/**
 * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
 * but performing a static call.
 *
 * _Available since v3.3._
 */
function functionStaticCall(
    address target,
    bytes memory data,
    string memory errorMessage
) internal view returns (bytes memory) {
    require(isContract(target), "Address: static call to non-contract");

    (bool success, bytes memory returndata) = target.staticcall(data);
    return verifyCallResult(success, returndata, errorMessage);
}

```

```

}

/**
 * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
 * but performing a delegate call.
 *
 * _Available since v3.4._
 */
function functionDelegateCall(address target, bytes memory data) internal returns
(bytes memory) {
    return functionDelegateCall(target, data, "Address: low-level delegate call
failed");
}

/**
 * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
 * but performing a delegate call.
 *
 * _Available since v3.4._
 */
function functionDelegateCall(
    address target,
    bytes memory data,
    string memory errorMessage
) internal returns (bytes memory) {
    require(isContract(target), "Address: delegate call to non-contract");

    (bool success, bytes memory returndata) = target.delegatecall(data);
    return verifyCallResult(success, returndata, errorMessage);
}

/**
 * @dev Tool to verifies that a low level call was successful, and revert if it
wasn't, either by bubbling the
 * revert reason using the provided one.
 *
 * _Available since v4.3._
 */
function verifyCallResult(
    bool success,
    bytes memory returndata,
    string memory errorMessage
) internal pure returns (bytes memory) {
    if (success) {

```



```

        return returndata;
    } else {
        // Look for revert reason and bubble it up if present
        if (returndata.length > 0) {
            // The easiest way to bubble the revert reason is using memory via
assembly

            assembly {
                let returndata_size := mload(returndata)
                revert(add(32, returndata), returndata_size)
            }
        } else {
            revert(errorMessage);
        }
    }
}

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your
contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)` , etc.
 */
library SafeERC20 {

    using Address for address;

    function safeTransfer(IERC20 token, address to, uint256 value ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to,
value));
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a
contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it
must not be false).
     * @param token The token targeted by the call.

```

```

    * @param data The call data (encoded using abi.encode or one of its variants).
    */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data
        size checking mechanism, since
        // we're implementing it ourselves. We use {Address.functionCall} to perform this
        call, which verifies that
        // the target address contains contract code and also asserts for success in the
        low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level
        call failed");
        if (returndata.length > 0) {
            // Return data is optional
            require(abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not
            succeed");
        }
    }
}

/*
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with GSN meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view returns (address payable) {
        return payable(msg.sender);
    }

    function _msgData() internal view returns (bytes memory) {
        this; // silence state mutability warning without generating bytecode - see
https://github.com/ethereum/solidity/issues/2691
        return msg.data;
    }
}

/**
 * @dev Contract module which provides a basic access control mechanism, where

```

```

* there is an account (an owner) that can be granted exclusive access to
* specific functions.
*
* By default, the owner account will be the one that deploys the contract. This
* can later be changed with {transferOwnership}.
*
* This module is used through inheritance. It will make available the modifier
* `onlyOwner`, which can be applied to your functions to restrict their use to
* the owner.
*/
abstract contract Ownable is Context {
    address private _owner;
    address private _newOwner;

    event OwnershipTransferred(address indexed previousOwner, address indexed
newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor () {
        address msgSender = _msgSender();
        _owner = msgSender;
        emit OwnershipTransferred(address(0), msgSender);
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Accept the ownership transfer. This is to make sure that the contract is
     * transferred to a working address

```

```

*
* Can only be called by the newly transfered owner.
*/
function acceptOwnership() public {
    require(_msgSender() == _newOwner, "Ownable: only new owner can accept
ownership");
    address oldOwner = _owner;
    _owner = _newOwner;
    _newOwner = address(0);
    emit OwnershipTransferred(oldOwner, _owner);
}

/**
 * @dev Transfers ownership of the contract to a new account (`newOwner`).
 *
 * Can only be called by the current owner.
 */
function transferOwnership(address newOwner) public onlyOwner {
    //SlowMist// This check is quite good in avoiding losing control of the contract
caused by user mistakes
    require(newOwner != address(0), "Ownable: new owner is the zero address");
    _newOwner = newOwner;
}
}

/**
 * @dev Contract module which allows children to implement an emergency stop
 * mechanism that can be triggered by an authorized account.
 *
 * This module is used through inheritance. It will make available the
 * modifiers `whenNotFrozen` and `whenFrozen`, which can be applied to
 * the functions of your contract. Note that they will not be pausable by
 * simply including this module, only once the modifiers are put in place.
 */
abstract contract Pausable is Context, Ownable {
    /**
     * @dev Emitted when the freeze is triggered by `account`.
     */
    event Frozen(address account);

    /**
     * @dev Emitted when the freeze is lifted by `account`.
     */
    event Unfrozen(address account);

```

```
bool private _frozen;

/**
 * @dev Initializes the contract in unfrozen state.
 */
constructor () {
    _frozen = false;
}

/**
 * @dev Returns true if the contract is frozen, and false otherwise.
 */
function frozen() public view returns (bool) {
    return _frozen;
}

/**
 * @dev Modifier to make a function callable only when the contract is not frozen.
 *
 * Requirements:
 *
 * - The contract must not be frozen.
 */
modifier whenNotFrozen() {
    require(!_frozen(), "Freezable: frozen");
    _;
}

/**
 * @dev Modifier to make a function callable only when the contract is frozen.
 *
 * Requirements:
 *
 * - The contract must be frozen.
 */
modifier whenFrozen() {
    require(_frozen(), "Freezable: frozen");
    _;
}

/**
 * @dev Triggers stopped state.
 */
```

```

* Requirements:
*
* - The contract must not be frozen.
*/
function _freeze() internal whenNotFrozen {
    _frozen = true;
    emit Frozen(_msgSender());
}

/**
 * @dev Returns to normal state.
 *
 * Requirements:
 *
 * - Can only be called by the current owner.
 * - The contract must be frozen.
 */
function _unfreeze() internal whenFrozen {
    _frozen = false;
    emit Unfrozen(_msgSender());
}
}

/**
 * @title Blacklistable Token
 * @dev Allows accounts to be blacklisted by a "blacklister" role
 */
contract Blacklistable is Context, Ownable {

    mapping(address => bool) internal _blacklisted;

    event Blacklisted(address indexed account_);
    event UnBlacklisted(address indexed account_);
    event BlacklisterChanged(address indexed newBlacklister_);

    /**
     * @dev Throws if argument account is blacklisted
     * @param account_ The address to check
     */
    modifier notBlacklisted(address account_) {
        require(!_blacklisted[account_], "Blacklistable: account is blacklisted");
        _;
    }
}

```

```

/**
 * @dev Checks if account is blacklisted
 * @param account_ The address to check
 */
function isBlacklisted(address account_) external view returns (bool) {
    return _blacklisted[account_];
}

/**
 * @dev Adds account to blacklist
 * @param account_ The address to blacklist
 */
function blacklist(address account_) external onlyOwner {
    _blacklisted[account_] = true;
    emit Blacklisted(account_);
}

/**
 * @dev Removes account from blacklist
 * @param account_ The address to remove from the blacklist
 */
function unBlacklist(address account_) external onlyOwner {
    _blacklisted[account_] = false;
    emit UnBlacklisted(account_);
}
}

/**
 * @dev Enable owner to withdraw tokens that wrongly sent to the contract
 */
contract Withdrawable is Ownable {
    using SafeERC20 for IERC20;

    /// @dev Emit `Withdrawn` when owner withdraw fund from the contract
    event Withdrawn(address indexed owner, address indexed recipient, address indexed
token, uint256 value);

    /// @dev withdraw token from contract
    /// @param token_ address of the token, use address(0) to withdraw gas token
    /// @param destination_ recipient address to receive the fund
    /// @param amount_ amount of fund to withdraw
    //Slowmist// The owner role can withdraw the any tokens from this contract to a
specific address
    function withdraw(address token_, address destination_, uint256 amount_) external

```

```

onlyOwner {
    require(destination_ != address(0), "Withdrawable: Destination is zero address");

    uint256 availableAmount;
    if(token_ == address(0)) {
        availableAmount = address(this).balance;
    } else {
        availableAmount = IERC20(token_).balanceOf(address(this));
    }

    require(amount_ <= availableAmount, "Withdrawable: Not enough balance");

    if(token_ == address(0)) {
        //SlowMist// There is no upper limit of the gas to execute the low-level call
to withdraw the native tokens
        destination_.call{value:amount_}("");
    } else {
        IERC20(token_).safeTransfer(destination_, amount_);
    }

    emit Withdrawn(_msgSender(), destination_, token_, amount_);
}

/// @dev withdraw NFT from contract
/// @param token_ address of the token, use address(0) to withdraw gas token
/// @param destination_ recipient address to receive the fund
/// @param tokenId_ ID of NFT to withdraw
//SlowMist// The owner role can withdraw the any NFT tokens from this contract to a
specific address
function withdrawNft(address token_, address destination_, uint256 tokenId_)
external onlyOwner {
    require(destination_ != address(0), "Withdrawable: destination is zero address");

    IERC721(token_).transferFrom(address(this), destination_, tokenId_);

    emit Withdrawn(_msgSender(), destination_, token_, 1);
}

/**
 * @dev Implementation of the {IERC20} interface.
 *
 * This implementation is agnostic to the way tokens are created. This means
 * that a supply mechanism has to be added in a derived contract using {_mint}.

```



```

* For a generic mechanism see {ERC20PresetMinterPauser}.
*
* TIP: For a detailed writeup see our guide
* https://forum.zeppelin.solutions/t/how-to-implement-erc20-supply-
mechanisms/226[How
* to implement supply mechanisms].
*
* We have followed general OpenZeppelin guidelines: functions revert instead
* of returning `false` on failure. This behavior is nonetheless conventional
* and does not conflict with the expectations of ERC20 applications.
*
* Additionally, an {Approval} event is emitted on calls to {transferFrom}.
* This allows applications to reconstruct the allowance for all accounts just
* by listening to said events. Other implementations of the EIP may not emit
* these events, as it isn't required by the specification.
*
* Finally, the non-standard {decreaseAllowance} and {increaseAllowance}
* functions have been added to mitigate the well-known issues around setting
* allowances. See {IERC20-approve}.
*/
contract CUSD is Context, Ownable, Pausable, Blacklistable, Withdrawable, IERC20 {
    using SafeMath for uint256;

    uint256 private _totalSupply;

    string private constant _name = "Coin98 Dollar";
    string private constant _symbol = "CUSD";
    uint8 private constant _decimals = 18;

    mapping (address => uint256) private _balances;

    mapping (address => mapping (address => uint256)) private _allowances;

    address private _minter;

    /**
     * @dev Sets the values for {name} and {symbol}, initializes {decimals} with
     * a default value of 18.
     *
     * To select a different value for {decimals}, use {_setupDecimals}.
     *
     * All three of these values are immutable: they can only be set once during
     * construction.
     */

```

```

constructor() {
    _minter = _msgSender();
}

/**
    @dev Emit when new minter is assigned
    */
event MinterUpdated(address indexed previousMinter, address indexed newMinter);

/**
    * @dev Returns the name of the token.
    */
function name() public view returns (string memory) {
    return _name;
}

/**
    * @dev Returns the symbol of the token, usually a shorter version of the
    * name.
    */
function symbol() public view returns (string memory) {
    return _symbol;
}

/**
    * @dev Returns the number of decimals used to get its user representation.
    * For example, if `decimals` equals `2`, a balance of `505` tokens should
    * be displayed to a user as `5,05` (`505 / 10 ** 2`).
    *
    * Tokens usually opt for a value of 18, imitating the relationship between
    * Ether and Wei. This is the value {ERC20} uses, unless {_setupDecimals} is
    * called.
    *
    * NOTE: This information is only used for _display_ purposes: it in
    * no way affects any of the arithmetic of the contract, including
    * {IERC20-balanceOf} and {IERC20-transfer}.
    */
function decimals() public view returns (uint8) {
    return _decimals;
}

/**
    * @dev See {IERC20-totalSupply}.
    */

```

```

function totalSupply() public view override returns (uint256) {
    return _totalSupply;
}

/**
 * @dev See {IERC20-balanceOf}.
 */
function balanceOf(address account) public view override returns (uint256) {
    return _balances[account];
}

/**
 * @dev Extra address which has privilege to mint more token
 */
function minter() public view returns (address) {
    return _minter;
}

/**
 * @dev Throws if called by any account other than the owner or minter.
 */
modifier onlyMinter() {
    require(owner() == _msgSender() || _minter == _msgSender(), "ERC20: caller is not minter");
    _;
}

/**
 * @dev See {IERC20-transfer}.
 *
 * Requirements:
 *
 * - `recipient` cannot be the zero address.
 * - the caller must have a balance of at least `amount`.
 */
function transfer(address recipient, uint256 amount) public override
    notBlacklisted(_msgSender())
    notBlacklisted(recipient)
    returns (bool)
{
    _transfer(_msgSender(), recipient, amount);
    //SlowMist// The return value conforms to the EIP20 specification
    return true;
}

```

```

/**
 * @dev See {IERC20-allowance}.
 */
function allowance(address owner, address spender) public view override returns
(uint256) {
    return _allowances[owner][spender];
}

/**
 * @dev See {IERC20-approve}.
 *
 * Requirements:
 *
 * - `spender` cannot be the zero address.
 */
function approve(address spender, uint256 amount) public override
    notBlacklisted(_msgSender())
    notBlacklisted(spender)
    returns (bool)
{
    _approve(_msgSender(), spender, amount);
    //SlowMist// The return value conforms to the EIP20 specification
    return true;
}

/**
 * @dev See {IERC20-transferFrom}.
 *
 * Emits an {Approval} event indicating the updated allowance. This is not
 * required by the EIP. See the note at the beginning of {ERC20}.
 *
 * Requirements:
 *
 * - `sender` and `recipient` cannot be the zero address.
 * - `sender` must have a balance of at least `amount`.
 * - the caller must have allowance for ``sender``'s tokens of at least
 *   `amount`.
 */
function transferFrom(address sender, address recipient, uint256 amount) public
override
    notBlacklisted(_msgSender())
    notBlacklisted(sender)
    notBlacklisted(recipient)

```

```

    returns (bool)
{
    _transfer(sender, recipient, amount);
    _approve(sender, _msgSender(), _allowances[sender][_msgSender()].sub(amount,
"ERC20: transfer amount exceeds allowance"));
    //SlowMist// The return value conforms to the EIP20 specification
    return true;
}

/**
 * @dev Atomically increases the allowance granted to `spender` by the caller.
 *
 * This is an alternative to {approve} that can be used as a mitigation for
 * problems described in {IERC20-approve}.
 *
 * Emits an {Approval} event indicating the updated allowance.
 *
 * Requirements:
 *
 * - `spender` cannot be the zero address.
 */
function increaseAllowance(address spender, uint256 addedValue) public
    notBlacklisted(_msgSender())
    notBlacklisted(spender)
    returns (bool)
{
    _approve(_msgSender(), spender, _allowances[_msgSender()]
[spender].add(addedValue));
    return true;
}

/**
 * @dev Atomically decreases the allowance granted to `spender` by the caller.
 *
 * This is an alternative to {approve} that can be used as a mitigation for
 * problems described in {IERC20-approve}.
 *
 * Emits an {Approval} event indicating the updated allowance.
 *
 * Requirements:
 *
 * - `spender` cannot be the zero address.
 * - `spender` must have allowance for the caller of at least
 * `subtractedValue`.

```

```

*/
function decreaseAllowance(address spender, uint256 subtractedValue) public
    notBlacklisted(_msgSender())
    notBlacklisted(spender)
    returns (bool)
{
    _approve(_msgSender(), spender, _allowances[_msgSender()]
[spender].sub(subtractedValue, "ERC20: decreased allowance below zero"));
    return true;
}

/**
 * @dev Issues `amount` tokens to the designated `address`.
 *
 * Can only be called by the current owner.
 * See {ERC20-_mint}.
 */
//SlowMist// The Minter role can mint tokens arbitrarily through the mint function
and there is no upper limit on the amount of tokens that can be minted
function mint(address account, uint256 amount) public
    onlyMinter
    notBlacklisted(_msgSender())
    notBlacklisted(account)
{
    _mint(account, amount);
}

/**
 * @dev Destroys `amount` tokens from the caller.
 *
 * See {ERC20-_burn}.
 */
function burn(uint256 amount) public
    notBlacklisted(_msgSender())
{
    _burn(_msgSender(), amount);
}

/**
 * @dev Disable the {transfer}, {mint} and {burn} functions of contract.
 *
 * Can only be called by the current owner.
 * The contract must not be frozen.
 */

```

```

function freeze() public onlyOwner {
    _freeze();
}

/**
 * @dev Enable the {transfer}, {mint} and {burn} functions of contract.
 *
 * Can only be called by the current owner.
 * The contract must be frozen.
 */
function unfreeze() public onlyOwner {
    _unfreeze();
}

/**
 * @dev Set new minter
 */
function setMinter(address newMinter) public onlyOwner {
    address oldMinter = _minter;
    _minter = newMinter;
    emit MinterUpdated(oldMinter, newMinter);
}

/**
 * @dev Moves tokens `amount` from `sender` to `recipient`.
 *
 * This is internal function is equivalent to {transfer}, and can be used to
 * e.g. implement automatic token fees, slashing mechanisms, etc.
 *
 * Emits a {Transfer} event.
 *
 * Requirements:
 *
 * - `sender` cannot be the zero address.
 * - `recipient` cannot be the zero address.
 * - `sender` must have a balance of at least `amount`.
 */
function _transfer(address sender, address recipient, uint256 amount) internal {
    require(sender != address(0), "ERC20: transfer from the zero address");
    //SlowMist// This kind of check is very good, avoiding user mistake leading to
the loss of token during transfer
    require(recipient != address(0), "ERC20: transfer to the zero address");

    _beforeTokenTransfer(sender, recipient, amount);

```

```

    _balances[sender] = _balances[sender].sub(amount, "ERC20: transfer amount exceeds
balance");
    _balances[recipient] = _balances[recipient].add(amount);
    emit Transfer(sender, recipient, amount);
}

/** @dev Creates `amount` tokens and assigns them to `account`, increasing
 * the total supply.
 *
 * Emits a {Transfer} event with `from` set to the zero address.
 *
 * Requirements:
 *
 * - `to` cannot be the zero address.
 */
function _mint(address account, uint256 amount) internal {
    require(account != address(0), "ERC20: mint to the zero address");

    _beforeTokenTransfer(address(0), account, amount);

    _totalSupply = _totalSupply.add(amount);
    _balances[account] = _balances[account].add(amount);
    emit Transfer(address(0), account, amount);
}

/**
 * @dev Destroys `amount` tokens from `account`, reducing the
 * total supply.
 *
 * Emits a {Transfer} event with `to` set to the zero address.
 *
 * Requirements:
 *
 * - `account` cannot be the zero address.
 * - `account` must have at least `amount` tokens.
 */
function _burn(address account, uint256 amount) internal {
    require(account != address(0), "ERC20: burn from the zero address");

    _beforeTokenTransfer(account, address(0), amount);

    _balances[account] = _balances[account].sub(amount, "ERC20: burn amount exceeds
balance");

```



```

    _totalSupply = _totalSupply.sub(amount);
    emit Transfer(account, address(0), amount);
}

/**
 * @dev Sets `amount` as the allowance of `spender` over the `owner` s tokens.
 *
 * This internal function is equivalent to `approve`, and can be used to
 * e.g. set automatic allowances for certain subsystems, etc.
 *
 * Emits an {Approval} event.
 *
 * Requirements:
 *
 * - `owner` cannot be the zero address.
 * - `spender` cannot be the zero address.
 */
function _approve(address owner, address spender, uint256 amount) internal {
    require(owner != address(0), "ERC20: approve from the zero address");
    //SlowMist// This kind of check is very good, avoiding user mistake leading to
approve errors
    require(spender != address(0), "ERC20: approve to the zero address");

    _allowances[owner][spender] = amount;
    emit Approval(owner, spender, amount);
}

/**
 * @dev Hook that is called before any transfer of tokens. This includes
 * minting and burning.
 *
 * Calling conditions:
 *
 * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
 * will be transferred to `to`.
 * - when `from` is zero, `amount` tokens will be minted for `to`.
 * - when `to` is zero, `amount` of ``from``'s tokens will be burned.
 * - `from` and `to` are never both zero.
 *
 * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-
hooks[Using Hooks].
 */
function _beforeTokenTransfer(address from, address to, uint256 amount) internal {
    require(!frozen(), "ERC20: token transfer while frozen");

```

```
}
```

```
function _afterTokenTransfer(  
    address from,  
    address to,  
    uint256 amount  
) internal {}  
}
```

Statement

SlowMist issues this report with reference to the facts that have occurred or existed before the issuance of this report, and only assumes corresponding responsibility based on these.

For the facts that occurred or existed after the issuance, SlowMist is not able to judge the security status of this project, and is not responsible for them. The security audit analysis and other contents of this report are based on the documents and materials provided to SlowMist by the information provider till the date of the insurance report (referred to as "provided information"). SlowMist assumes: The information provided is not missing, tampered with, deleted or concealed. If the information provided is missing, tampered with, deleted, concealed, or inconsistent with the actual situation, the SlowMist shall not be liable for any loss or adverse effect resulting therefrom. SlowMist only conducts the agreed security audit on the security situation of the project and issues this report. SlowMist is not responsible for the background and other conditions of the project.



Official Website
www.slowmist.com



E-mail
team@slowmist.com



Twitter
[@SlowMist_Team](https://twitter.com/SlowMist_Team)



Github
<https://github.com/slowmist>